C1M5

Tangents and Velocity

Suppose that we operate an emergency vehicle on a busy highway and that the hospital that serves this community is on the highway 10 miles to the east of our base. An efficiency expert is hired to record our location electronically every minute for a week and it is determined that the hospital is located at $+10$ miles. The clock is started at midnight and at 12:03 AM a call from an accident two miles to the west comes in. We race to the scene, arriving at 12:06, put an injured patient in the vehicle, and head for the hospital 4 minutes later with a police escort, slowing somewhat for an intersection at the $+5$ mile mark. We reach the hospital at 12:19 AM.

Later, we are asked what our average speed was going to the hospital from the accident scene. Our efficiency expert looks at the data and points out that we went 12 miles in 9 minutes, so our average speed was $\dfrac{12}{9/60} = 80$ mph. Then we are asked if we ever exceeded 80 mph, which is our maximum emergency speed allowed by local statutes. Knowing that there is "ramp" time when we accelerate and decelerate, and that we had slowed to $50 - 60$ mph at that intersection, I concluded that we must have exceeded 80 mph at some time in the rush to the hospital. In fact, having glanced at the speedometer a couple of times, I knew that we had almost hit 90 mph once. After I admitted that we had exceeded 80 mph, the efficiency expert smiled knowingly and said,"At 12:12 you were at mile 1.6, and at 12:13 you were at mile 3.05, so your speed for that minute was $\dfrac{3.05-1.6}{1/60} = (1.45)\cdot 60 = 87$ mph."

I asked the expert if I could have the data and it was provided, so I decided to use Maple to analyze the situation. I began by listing the mile information and then I associated a time in minutes with each location, starting with 0. This generated a sequence, but by enclosing it in square brackets, it became a list. Then I plotted the data as a line graph and as a point graph, putting the plots on the same coordinate axes.

```
> restart:    with(plots):
> datapoints:=[0,0,0,0,-.3,-1.2,-2,-2,-2,-2,-1,.2,1.6,3.05,4.4,5.5,6.8,8.1,9.2,10];
```

$$datapoints := [0, 0, 0, 0, -.3, -1.2, -2, -2, -2, -2, -1, .2, 1.6, 3.05, 4.4, 5.5, 6.8, 8.1, 9.2, 10]$$

```
> nops(datapoints);
```

$$20$$

The command `nops` returns the number of parts of its argument. This told me that I had entered 20 values.

```
> seq1:=[seq([i,datapoints[i+1]],i=0..19)];
```

$$seq1 := [[0,0],[1,0],[2,0],[3,0],[4,-.3],[5,-1.2],[6,-2],[7,-2],[8,-2],[9,-2],[10,-1],[11,.2],[12,1.6],$$
$$[13,3.05],[14,4.4],[15,5.5],[16,6.8],[17,8.1],[18,9.2],[19,10]]$$

```
> A:=pointplot(seq1,style=line,color=red,scaling=constrained):
> B:=pointplot(seq1,style=point,symbol=circle,symbolsize=12,color=blue):
> display(A,B);
```



On the vertical axis we have the mile mark, and on the horizontal axis we have the time scale in minutes.

I decided to compute the average velocity for each minute. By subtracting the old mile mark from the new one, I got the net change in distance. Then I divided that by the change in time, which is $1/60^{\text{th}}$ of an hour, to get miles per hour.

```
> seq2:=seq((datapoints[i+1]-datapoints[i])/(1/60),i=1..19);
```

$$seq2 := 0, 0, 0, -18.0, -54.0, -48.0, 0, 0, 0, 60, 72.0, 84.0, 87.00, 81.00, 66.0, 78.0, 78.0, 66.0, 48.0$$

As I compared the speeds listed, I looked at the graph and noted that these numbers were just the slopes of the lines that joined the data points on the graph. Sure enough, our top speed listed was 87 mph and the slope of the line between 12 and 13 was the steepest of any of the slopes. I also realized that because we initially went west, the speeds then are shown as negative numbers. And, when we were stationary our speed was 0.

If the expert had been able to give data that was recorded every second, then it would have been overwhelming in size, but the speeds calculated would have been very accurate, and the graph would have been almost smooth. And, the accuracy of the computations of speed would have made the evidence against us for speeding also overwhelming.

In this discussion the terms ***velocity*** and ***speed*** have been used as if they are interchangeable. This is misleading because they are related, but different. Think of speed as the absolute value of velocity. In our computations we determined the average velocity over a one minute time span. Speedometers record speed, not velocity, and they allow only for non-negative values. In addition to how fast the vehicle was moving, there was also a direction involved. Motion to the east produced a positive velocity, while motion towards the west yielded negative values. We chose east as our positive direction and recorded ***signed values***, i.e. $+, -$ from the base, not just distance from that point. We did not really answer the question of how fast the vehicle was moving at some instant in time, rather we computed a sequence of average velocities over one minute intervals. As you should expect, in order to approximate the ***instantaneous velocity*** we would need to compute the average velocity over shorter and shorter time intervals. If this 'seems like ***déjà vu*** all over again', then you are right. Computing an instantaneous velocity and computing the slope of a tangent line are the same processes. Suppose that the dependent (y value) variable is distance. When the independent (x value) variable is distance we get a slope. When the independent variable is regarded as time (x value, but think of it as t), then our computation is a velocity.

What was our average velocity over the 19 minutes? The net change in distance was 10 miles, while the change in time was 19/60 hours. This yields $10/(19/60)) = \frac{600}{19} \approx 31.6$ mph. There were probabably two instants in time when our instantaneous velocity was $\frac{600}{19}$, once as we accelerated away from the accident and again as we decelerated as we approached the hospital. So this average value over a long time interval doesn't indicate our instantaneous velocity over this time period very well. To get a good approximation for an instantaneous velocity, we must use smaller and smaller time intervals. We are clearly dancing around a concept called a ***limit***. In mathematics, whenever a limit is discussed there are always two ingredients involved, ***accuracy*** and ***control***. In the section on limits we will be more precise, but in order to get a good approximation for our instantaneous velocity we 'invoke more control', i.e. we use smaller and smaller time intervals.

Slope

When calculating an approximation for the slope of the tangent line for $y = a^x$ at $x = 0$ we used a symmetric form of slope by taking a point on each side of 0. It might be helpful if you see that our approach is consistent with your prior experience. At x_0, you took an increment h which should be regarded as a small number, either positive or negative, but we almost always draw the positive case. If $h > 0$ then on the x-axis the values $-h$ and h are $2h$ units apart. The usual form for the slope of a secant line of $y = f(x)$ at $x = x_0$ is given by

$$\frac{f(x_0+h)-f(x_0)}{h} \quad \text{on the right} \qquad \text{and} \qquad \frac{f(x_0-h)-f(x_0)}{-h} \quad \text{on the left}$$

and then we take the positive h smaller and smaller. Consider what happens when we insert 0 in the symmetric form in a clever way.

$$\begin{aligned}\frac{f(x_0+h)-f(x_0-h)}{2h} &= \frac{f(x_0+h)-f(x_0)+f(x_0)-f(x_0-h)}{2h} \\ &= \frac{1}{2}\left(\frac{f(x_0+h)-f(x_0)}{h} + \frac{f(x_0-h)-f(x)}{-h}\right)\end{aligned}$$

So using the symmetric form is really just taking the average of the right and left forms. When our function is smooth, a term that we will define more precisely later, we will get the same approximation either way. In fact, we will find our approximation by using only the form

$$\frac{f(x_0+h)-f(x_0)}{h}$$

and will permit h to be both positive and negative as it becomes small.

Maple Example: Use Maple to find an equation for the line tangent to $y = \sqrt{x}$ at $x = 2$. Plot the line and the function on the same coordinate axes.

```
> restart:    with(plots):
> f:=x->sqrt(x);  x0:=2;  y0:=f(x0);
```

$$f := \mathrm{sqrt}$$
$$x0 := 2$$
$$y0 := \sqrt{2}$$

```
> F:=h->(f(x0+h)-f(x0))/h;
```

$$F := h \longrightarrow \frac{f(x0+h) - f(x0)}{h}$$

```
> S1:=seq(evalf(F(1/2^i)),i=6..12);
```

$$S1 := .35286555, .3532089, .3533810, .3534673, .353511, .353533, .353544$$

```
> S2:=seq(evalf(F(-1/2^i)),i=6..12);
```

$$S2 := .35424661, .3538993, .3537261, .3536395, .353596, .353574, .353563$$

```
> abs(S1[7]-S2[7]);
```

$$.000019$$

```
> m:=evalf(S1[7],4);
```

$$m := .3535$$

```
> eq1:=y-y0=m*(x-x0);
```

$$eq1 := y - \sqrt{(2)} = .3535 * x - .7070$$

```
> y:=solve(eq1,y);
```

$$y := .7072135624 + .3535000000\,x$$

```
> P:=plot(y,x=0..4,color=blue):
> Q:=plot(f(x),x=0..4,color=red):
> display(P,Q);
```



Honesty requires that I show you an easy way to plot a function and its tangent line at a specified point using Maple. In the package `student` there is a command `showtangent` that does the job cleanly and without fuss. For example,

```
> restart:    with(student):
> showtangent(x*sin(Pi*x),x=.6,x=0..2,color=blue);
```



C1M5 Problems: Use Maple to solve the problems and plot the graphs.

1. Our efficiency expert from the text moved on to a different ambulance location and set up the same program. The data below was collected at one minute intervals and the first value is for 2 AM. The last value is at the hospital. Provide as much information about the ambulance run as possible, including the average velocities.

$$data = [0, 0, .3, 1.4, 2.7, 4, 4.8, 5.1, 5.1, 5.1, 5.1, 6, 7.3, 8.7, 10, 11, 11.7, 12.2, 12.6]$$

2. For $y = x\ \ln(x)$ and $x_0 = 3$, find an equation for the line tangent to the curve at x_0 and plot the graphs of the function and the line on the same coordinate axes.

3. For $y = x\,2^{-x}$ and $x_0 = \frac{-3}{2}$, find an equation for the line tangent to the curve at x_0 and plot the graphs of the function and the line on the same coordinate axes.

4. For $y = \sinh(x)$ and $x_0 = 2$, find an equation for the line tangent to the curve at x_0 and plot the graphs of the function and the line on the same coordinate axes.

Warning! It is perfectly reasonable to do Problem 2 and then, when doing 3 and 4, copy, paste it below, and change the values and execute. If you do, some values will carry over that you do not wish for Maple to remember. It is suggested that you insert a command line between the problems:

```
> restart:    with(plots):
```